Math 140 Analyzing Real Sheep Data
Today we’ll work with a dataset from the “Selfish-herd behaviour of sheep under threat” article by King et al. (2012), Current Biology 22:R561-2. We’ll see how to view and analyze the data within NetLogo, but you may also use other platforms like Excel or Matlab or R that you like to use. I have compiled the data into a space-separated text file “SheepData08.txt”. The 1st row has time in seconds, the 2nd and 3rd rows are the dog’s x and y coordinates in meters, then each successive pair of rows corresponds to a sheep.
1. Start a new NetLogo file. Change the settings to have the origin in the lower left corner and max coordinates of 180 for both x and y. Choose a patch size so the world window is a good size.
2. We will read in the file line by line into lists of x- and y-coordinates for the sheepdog and the 46 sheep. The time is in seconds and is the same for all, so we’ll make the time list “tlist” a global. Each dog and sheep will own their xlist and ylist.
globals [ tlist ]

breed [ sheep a_sheep ]

breed [ dogs dog ]

sheep-own [ xlist ylist ]

dogs-own [xlist ylist ]

to setup

clear-all

ask patches

  [ set pcolor lime ]

  ifelse ( file-exists? "SheepData08.txt" ) [ ;; make sure the file exists first

    file-open "SheepData08.txt"   ;; open the file, so we can read it.

    set tlist read-from-string (word "[" file-read-line "]" ) ;; read a line of the file and convert to list
create-dogs 1 ;; one sheepdog
    [ set shape "wolf"

      set color brown

      set size 8

      set xlist read-from-string (word "[" file-read-line "]" )

      set ylist read-from-string (word "[" file-read-line "]" )

      setxy item 0 xlist item 0 ylist] ; NetLogo starts indexing of lists at 0

 while [ not file-at-end? ] ;; read in sheep coordinates until reach end of file
    [ create-sheep 1

      [ set shape "sheep"

        set size 4

        set color random 100 / 10 ;; sheep are colored range of grays
        set xlist read-from-string (word "[" file-read-line "]" )

        set ylist read-from-string (word "[" file-read-line "]" )

        setxy item 0 xlist item 0 ylist

        ] ]

    file-close ] ;; Done reading in coordinate information.  Close the file.

 [ user-message "There is no SheepData08.txt file in current directory!" ];; reports error if no file

  reset-ticks

end

3. Create a setup button and verify that the setup procedure runs correctly. You can speed up the initialization by sliding the speed up to “faster” at the top of the interface. If you have file errors, you may need to type file-close at the bottom of the interface at the observer prompt to reset the file reading.

4. Create a go button and add the following to the code. The dog and sheep will move at each tick as indicated by the data. You should now slide the speed of the interface to “slower” or it will go past in the blink of an eye.
to go

  ifelse ticks < length tlist [

  ask dogs

  [ setxy item ticks xlist item ticks ylist ]

  ask sheep

  [ setxy item ticks xlist item ticks ylist ] ]

  [ stop ]

  tick

end
5. Carefully examine the behavior of the sheep and the dog. How do the sheep react to the dog? How close does he need to be to trigger the sheep (the “panic distance”)? Do all of the sheep react at the same time? Does there appear to be a leader or do the sheep move around within the herd? Can you distinguish what direction the sheep move in relation to the dog and to the rest of the herd? 

6. We want to gather ideas to test through simulations to reproduce the sheep and dog behaviors. To accomplish this, we must compare outcomes of the real data to those of simulations. Add the nearest neighbor (NN1) code to this NetLogo file to see what how the NN1 distribution changes as the dog runs toward the sheep and as he herds them. Also add the polarization (“synchrony”) measure. 
